The Mutilated Checkerboard Problem in the Lightweight Set Theory of Mizar

Piotr Rudnicki Department of Computing Science University of Alberta Edmonton, Alberta, Canada T6J 2H1

    piotr@cs.ualberta.ca
http://www.ualberta.ca/~piotr

November 20, 1995

Abstract

An 8�8 checkerboard with two diagonally opposite squares removed cannot be covered by 2�1 dominoes. John McCarthy postulates that a mechanized system of heavy duty set theory should accept his formal description of the proposition and his proof that the covering is impossible, as presented at the qed meeting in Warsaw in 1995. This note is a report on solving the mutilated checkerboard problem in the Mizar1 proof checking environment.

The presentation and solution of the mutilated checkerboard problem as suggested by John McCarthy is available from the qed workshop proceedings [4] and also from his home page^2. G. Bancerek, during the qed workshop, formalized the mutilated checkerboard problem in Mizar and proved the impossibility of the covering. An abstract of his solution is contained in [1]. The solution of the checkerboard problem that is discussed below has been developed independently of Bancerek’s work.

Appendix A contains the complete Mizar text of our solution of the mutilated checkerboard problem. Below, we first present how the formal description of the problem as proposed by J. McCarthy is expressed in Mizar without any essential changes. Then, we discuss a proof accepted by Mizar

1For information about Mizar visit http://www.cs.ualberta.ca/~piotr/Mizar 2http://www-formal.stanford.edu/jmc

of the impossibility of the mutilated checkerboard covering, in which we try to follow the the steps from the J. McCarthy’s note.

1 The formal description

The definitions used by J. McCarthy are expressible in Mizar with only minor changes, mainly due to different notation.

The boards that McCarthy defines as: Board = Z8 × Z8, (1) mutilated-board = Board − {(0, 0), (7, 7)}, (2) are introduced in the Mizar text as:

set Board = [:Segm 8, Segm 8:];
set Corners = {[0,0], [7,7]};
set mutilated Board = Board \ Corners;

The name for Corners is convenient as we will have to prove some facts about them. Also, the different spelling of mutilated-board in Mizar is because of convenience and not necessity.

J. McCarthy defines dominoes as follows: domino-on-board(x) ≡ (x ⊂ Board) ∧ card(x) = 2 ∧(∀x1 x2)(x1 ≠ x2 ∧ x1 ∈ x ∧ x2 ∈ x ⊃ adjacent(x1, x2)) (3) and then gives two alternative definitions for adjacent, labelling the second one as slightly tricky and expecting that it might be less convenient for automated provers: adjacent(x1, x2) ≡ |c(x1, 1) − c(x2, 1)| = 1 ∧c(x1, 2) = c(x2, 2) ∨|c(x1, 2) − c(x2, 2)| = 1 ∧ c(x1, 1) = c(x2, 1), (4)

adjacent(x1, x2) ≡ |c(x1, 1) − c(x2, 1)| + |c(x1, 2) − c(x2, 2)| = 1 (5) In Mizar we have to define things before using them. Therefore the definition of adjacent must precede the definition of domino on board. We have chosen the second definition of adjacent because it is more compact and did not cause difficulties in proving by hand.

definition let x1, x2 be Element of [:NAT,NAT:];
  pred x1 is-adjacent-to x2 means
:Adjacent Def: |.x1‘1-x2‘1.|+|.x1‘2-x2‘2.| = 1;
end;

I have slightly changed the definition of domino-on-board; it is more general but does not affect the proof in any substantial way.

definition let x be set;
  pred x is domino-on-grid means
:Domino Def: x c= [:NAT,NAT:] & Card x = 2
             & for x1, x2 being Element of [:NAT,NAT:]
                 st x1 <> x2 & x1 ∈ x & x2 ∈ x holds x1 is-adjacent-to x2;
end;

McCarthy’s definition of partial covering by dominoes: partial-covering(z) ≡ (∀x)(x ∈ z ⊃ domino-on-board(x)) ∧(∀x y)(x ∈ z ∧ y ∈ z ⊃ x = y ∨ x ∩ y = {}) (6)

is virtually identical in Mizar:

definition let z be set;
pred z is partial-covering means
:Part cover Def:
  (for x being set st x ∈ z holds x is domino-on-grid) &
   for x, y being set st x ∈ z & y ∈ z holds x = y or x ∩ y = Œ;
end;

2 Proofs

J. McCarthy claims that the proof as presented in Figure 1 should be admitted in any heavy duty set theory.

At the notational level, this proof can be almost literally rewritten into Mizar as presented in Figure 2.

One notational difference concerns the usage of the functor color that J. McCarthy introduces inside his proof. In Mizar, the functor color must be defined outside of the proof.

Theorem: ¬(∃z)(partial-covering(z) ∧ [ z = mutilated-board) (7) Proof: We define x ∈ Board ⊃ color(x) = rem(c(x, 1) + c(x, 2), 2) (8) domino-on-board(x) ⊃ (∃u v)(u ∈ x ∧ v ∈ x ∧ color(u) = 0 ∧ color(v) = 1), (9) partial-covering(z) ⊃ card({u ∈ S z|color(u) = 0}) = card({u ∈ S z|color(u) = 1}), (10)

card({u ∈ mutilated-board|color(u) = 0}) ≠ card({u ∈ mutilated-board|color(u) = 1}), (11)

and finally ¬(∃z)(partial-covering(z) ∧ mutilated-board = [ z) (12) Q.E.D.

Figure 1: McCarthy’s proof

The definition (8) has been generalized and is defined for all pairs of natural numbers.

definition let x be Element of [:NAT, NAT:];
  func color x -> Nat means
:color Def: it = x‘1+x‘2 mod 2;
  correctness;
end;

Unfortunately, the current version of Mizar (5.2.12) does not accept the proof from Figure 2 as is: all the intermediate steps have to be proven. The complete Mizar text of the proof is attached in appendix A. Here we will try to comment on its length and structure.

5 P. Rudnicki

theorem
       not ex z st z is partial-covering & union z = mutilated Board
proof
       given s such that
A1: s is partial-covering & union s = mutilated Board;
Domino: :: (9)
  x is domino-on-grid implies
          ex u, v st x = {u,v} & color u = 0 & color v = 1;
::> *4
Part cover: :: (10)
   z is partial-covering implies
              Card {u : u ∈ union z & color u = 0}
              = Card {v : v ∈ union z & color v = 1} by Domino;
::> *4
Mutilated Board: :: (11)
      Card {u : u ∈ mutilated Board & color u = 0}
      <> Card {v : v ∈ mutilated Board & color v = 1};
::> *4
 thus contradiction by A1, Part cover, Mutilated Board;
end;

::> 4: This inference is not accepted

Figure 2: McCarthy’s proof is not accepted by Mizar

Proving Domino

The proof of Domino takes 30 lines of Mizar source text and is a straightforward reasoning from the definitions: Domino Def, Adjacent Def, and

color Def. The Mizar data base article GOBOARD1 [3] contains a theorem

which states the equivalence between McCarthy’s two definitions of adjacency. For this reason, the second one has been used.

Proving Part cover

The proof of Part cover takes 60 lines. In order to show that two sets have equal cardinalities we first demonstrate that they are equipollent and then use the fact CARD 1:21 [2]. The equipollence of the white and the black squares on dominoes in a partial-covering named z is expressed as:

A2: {u : u ∈ union z & color u = 0} ß {v : v ∈ union z & color v = 1}

and its proof proceeds by definitional expansion of TARSKI:def 6 from [5]:

 definition let X,Y;
  pred X ß Y means
:: TARSKI: def 6
    ex Z st
       (for x st x ∈ X ex y st y ∈ Y & [x,y] ∈ Z) &
       (for y st y ∈ Y ex x st x ∈ X & [x,y] ∈ Z) &
        for x,y,z,u st [x,y] ∈ Z & [z,u] ∈ Z holds x = z iff y = u;
 end;

(In this definition all variables are typed as set.) In the proof, we take the following set of ordered pairs which establishes a mapping between the white and the black squares in z:

Z = {[u,v] : {u,v}∈z & color u = 0};

Then we prove that: Z is total on the white squares of z (15 lines), Z is onto the black squares of z (15 lines), and Z is a functional set, which completes the proof of the definiens of TARSKI:def 6.

Proving the Mutilated Board

The arguments leading to Mutilated Board are preceded by two lemmas named Board 01 and Corners 0, together occupying 160 lines of Mizar text. The following abbreviations are used:

set B0 = {u : u ∈ Board & color u = 0};
set B1 = {u : u ∈ Board & color u = 1};
set MB0 = {u : u ∈ mutilated Board & color u = 0};
set MB1 = {u : u ∈ mutilated Board & color u = 1};

for the sets of white and black squares in the complete and in the mutilated board, respectively.

The proof of the equipollence of white and black squares in the complete board:

Board 01: B0 ß B1

takes 75 lines and, like the proof of Part cover, proceeds by definitional expansion of TARSKI:def 6. The following set:

take Z = {[w,b] where w, b is Element of Board
                : color w = 0 & b=[w‘1,w‘2+1 mod 8]};

is used to establish the one-to-one mapping between the white and the black squares in the complete Board.

In the process of proving the lemma, we added several lemmas about the mod functor (we could not find anything appropriate in the Mizar data base). The auxiliary lemmas are placed at the beginning of the text in Appendix A and they occupy 55 lines together with their proofs. The auxiliary lemmas are not counted in the length of the proof of Board 01 as they state general facts about the mod functor. The Mizar data base is under development and sometimes even simple facts are not found in it.

That the removed corners are white has been expressed as:

Corners 0: Corners c= B0

and was proven separately (14 lines) since it is referenced two times later.

The demonstration that:

Mutilated Board: Card {u : u ∈ mutilated Board & color u = 0}
                 <> Card {v : v ∈ mutilated Board & color v = 1}

uses the two lemmas just proved and is almost 60 lines long. The argument is based on counting cardinalities of finite sets. First, in a sequence of steps we prove that all the sets of interest are finite:

reconsider Corners as finite set;
 . . .
reconsider Board as finite set;
 . . .
reconsider B0 as finite set;
 . . .
reconsider MB0 as finite set;

such that later we can do natural arithmetic on their cardinalities. After proving that:

MB0 = B0 \ Corners

we obtain:

A1: card MB0 = card B0 - card Corners by Corners 0, CARD 2:63;

from which we have that card MB0 <> card B0 and consequently

A2: not MB0 ß B0 by CARD 1:21;

And now, after showing MB1 = B1, we finally have:

not MB0 ß MB1 by Board 01, A2, WELLORD2:22;

with Mutilated Board just restating this fact in terms of cardinalities.

With all the lemmas proven, the proof of the main theorem can now be written as succinctly as J. McCarthy desires:

theorem
       not ex z st z is partial-covering & union z = mutilated Board
proof
       given z such that
A1: z is partial-covering & union z = mutilated Board;
A2: Card {u : u ∈ union z & color u = 0}
    = Card {v : v ∈ union z & color v = 1} by A1, Part cover;
A3: Card {u : u ∈ mutilated Board & color u = 0}
    <> Card {v : v ∈ mutilated Board & color v = 1} by Mutilated Board;
 thus contradiction by A1, A2, A3;
end;

References

[1] G. Bancerek. The mutilated chessboard problem — checker by Mizar.

In R. Matuszewski, editor, The QED Workshop II, pages 43–45. Warsaw University at Białystok, 1995. http://www.mcs.anl.gov/qed.

[2] Grzegorz Bancerek. Cardinal numbers. Formalized Mathematics, 1(2):377–382, 1990.

[3] Jarosław Kotowicz and Yatsuka Nakamura. Introduction to Go-board - part I. Formalized Mathematics, 3(1):107–115, 1992.

[4] J. McCarthy. The mutilated checkerboard is set theory. In R. Matuszewski, editor, The QED Workshop II, pages 25–26. Warsaw University at Białystok, 1995. http://www.mcs.anl.gov/qed.

[5] Andrzej Trybulec. Tarski Grothendieck set theory. Formalized Mathematics, 1(1):9–11, 1990.

9 P. Rudnicki

A The Mizar text

environ
 vocabulary FAM OP, EQUI REL, BOOLE, CARDINAL, COORD, NAT 1, GROUP CY, GRAPH,
             ANAL, REAL 1, MUTCHECK;
 clusters FINSET 1, GROUP 2;
 signature ARYTM, TARSKI, BOOLE, MCART 1, DOMAIN 1, FINSET 1, REAL 1,
             ANAL 1, NAT 1, WELLORD2, CARD 1, GR CY 1;
 requirements ARYTM;
 theorems TARSKI, AXIOMS, BOOLE, REAL 1, REAL 2, NAT 1, WELLORD2,
             CARD 1, CARD 2, CARD 4, SGRAPH1, GOBOARD1, GR CY 1, GROUP 4,
             ZFMISC 1, MCART 1;
 definitions TARSKI, BOOLE;
begin

:: Auxiliary facts ::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

reserve a, b, c, d for Nat;

Mod 1: 0<c implies ((a+b) mod c = (a+d) mod c iff b mod c = d mod c)
proof assume
A1: 0<c;
  hereby assume
      B1: (a+b) mod c = (a+d) mod c;
            consider n being Nat such that
      B2: a+b = c · n + (a+b mod c) & (a+b mod c) < c by A1, NAT 1:def 2;
            consider p being Nat such that
      B3: a+d = c · p + (a+d mod c) & (a+d mod c) < c by A1, NAT 1:def 2;
          a+(d-c · p) = a+d-c · p by REAL 1:17 .= a+b mod c by B1, B3, REAL 2:11
           .= a+b-c · n by B2, REAL 2:11 .= a+(b-c · n) by REAL 1:17; then
      B4: b-c · n = d-c · p by REAL 1:10;
    thus b mod c = b+c · p mod c by A1, GR CY 1:1 .= d+c · n mod c by B4, REAL 2:14
                .= d mod c by A1, GR CY 1:1;
  end; assume
A2: b mod c = d mod c;
      consider n being Nat such that
A3: b = c · n + (b mod c) & (b mod c) < c by A1, NAT 1:def 2;
      consider p being Nat such that
A4: d = c · p + (d mod c) & (d mod c) < c by A1, NAT 1:def 2;
A5: a+b-c · n = a+(b-c · n) by REAL 2:18 .= a+(b mod c) by A3, REAL 2:11
              .= a+(d-c · p) by A2, A4, REAL 2:11 .= a+d-c · p by REAL 2:18;
  thus a+b mod c = (a+b+c · p) mod c by A1, GR CY 1:1
        .= (a+d+c· n) mod c by A5, REAL 2:14 .= a+d mod c by A1, GR CY 1:1;
end;

Mod 2: 0<b & 0<c implies (a mod b · c) mod c = a mod c
proof assume
A1: 0<b & 0<c;
    set x = b · c; 0<>x by A1, REAL 1:23; then
      consider n being Nat such that
A2: a = x · n + (a mod x) & (a mod x) < x by NAT 1:def 2;
  thus (a mod x) mod c = ((a mod x)+ c · (b · n)) mod c by A1, GR CY 1:1
        .= ((a mod x)+c · b · n) mod c by AXIOMS:16 .= a mod c by A2;
end;

Mutilated Checkerboard in Mizar 10

Mod 3: 0<c & a<c & b<c & a<>b implies a mod c <> b mod c
proof assume
    A1: 0<c & a<c & b<c & a<>b; then a mod c = a & b mod c = b by GR CY 1:4;
  hence thesis by A1;
end;

Mod 4: 0<c & 0<b & b<c implies a mod c <> a+b mod c
proof assume
    A1: 0<c & 0<b & b<c & a mod c = a+b mod c; then
        a+0 mod c = a+b mod c; then 0 mod c = b mod c by A1, Mod 1;
  hence contradiction by A1, Mod 3;
end;

Aux 1: 0 <> 1;

definition let x be Element of [:NAT,NAT:];
 redefine func x‘1 -> Nat;
  coherence proof x‘1 is Element of NAT; hence thesis; end;
 redefine func x‘2 -> Nat;
  coherence proof x‘2 is Element of NAT; hence thesis; end;
end;

:: Problem description ::::::::::::::::::::::::::::::::::::::::::::::::::::::

set Board = [:Segm 8, Segm 8:];

set Corners = {[0,0], [7,7]};

set mutilated Board = Board \ Corners;

definition let x1, x2 be Element of [:NAT,NAT:];
  pred x1 is-adjacent-to x2 means
:Adjacent Def: |.x1‘1-x2‘1.|+|.x1‘2-x2‘2.| = 1;
end;

definition let x be set;
  pred x is domino-on-grid means
:Domino Def: x c= [:NAT,NAT:] & Card x = 2
             & for x1, x2 being Element of [:NAT,NAT:]
                 st x1 <> x2 & x1 ∈ x & x2 ∈ x holds x1 is-adjacent-to x2;
end;

definition let z be set;
pred z is partial-covering means
:Part cover Def:
  (for x being set st x ∈ z holds x is domino-on-grid) &
   for x, y being set st x ∈ z & y ∈ z holds x = y or x ∩ y = Œ;
end;

definition let x be Element of [:NAT, NAT:];
  func color x -> Nat means
:color Def: it = x‘1+x‘2 mod 2;
  correctness;
end;

11 P. Rudnicki

:: Proofs ::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

reserve x, y, z for set;
reserve u, v, w, b for Element of [:NAT,NAT:];

Domino: ::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::
x is domino-on-grid implies ex u, v st x = {u, v} & color u = 0 & color v = 1
proof assume x is domino-on-grid; then
A1: x c= [:NAT, NAT:] & Card x = 2
    & for w, b being Element of [:NAT,NAT:]
        st w<>b & w∈x & b∈x holds w is-adjacent-to b by Domino Def;
             2 is finite by CARD 1:69; then x is finite by A1, CARD 4:1; then
    consider w,b being set such that
A2: w∈x & b∈x & w<>b & x={w,b} by A1, SGRAPH1:8;
    reconsider w, b as Element of [:NAT,NAT:] by A1, A2;
             w is-adjacent-to b by A1, A2; then
             |.w‘1-b‘1.|+|.w‘2-b‘2.| = 1 by Adjacent Def; then
A3: |.w‘1-b‘1.| = 1 & w‘2=b‘2 or
               |.w‘2-b‘2.| = 1 & w‘1= b‘1 by GOBOARD1:2;
A4: color w = (w‘1+w‘2) mod 2 & color b = (b‘1+b‘2) mod 2 by color Def;
A5: now assume B1: color w = color b;
        per cases by A3, GOBOARD1:1;
        suppose C1: (w‘1 = b‘1 +1 or b‘1 = w‘1 +1) & (w‘2 = b‘2); then
                     w‘1 mod 2 = b‘1 mod 2 by B1, A4, Mod 1;
          hence contradiction by C1, Mod 4;
        suppose C1: (w‘2 = b‘2 +1 or b‘2 = w‘2 +1) & (w‘1 = b‘1); then
                     w‘2 mod 2 = b‘2 mod 2 by B1, A4, Mod 1;
          hence contradiction by C1, Mod 4;
    end;
 per cases by A4, GROUP 4:100;
 suppose B1: color w = 0; take w, b;
   thus thesis by A2, B1, A4, A5, GROUP 4:100;
 suppose B1: color w = 1; take b, w;
   thus thesis by A2, B1, A4, A5, GROUP 4:100;
end;

Part cover: ::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::
z is partial-covering implies Card {u : u ∈ union z & color u = 0}
                             = Card {v : v ∈ union z & color v = 1}
proof assume
A1: z is partial-covering;
A2: {u : u ∈ union z & color u = 0} ß {v : v ∈ union z & color v = 1}
    proof
          set W = {u : u∈union z & color u = 0};
          set B = {u : u∈union z & color u = 1};
          set Z = {[u,v] : {u,v}∈z & color u = 0};
     take Z;
     hereby :: Z is total on W
       let x be set; assume x∈W; then
             consider w such that
         C1: x = w & w∈union z & color w = 0;
             consider d being set such that
         C2: w∈d & d∈z by C1, TARSKI:def 4;
         C3: d is domino-on-grid by A1, C2, Part cover Def;

Mutilated Checkerboard in Mizar 12

        consider dw, db being Element of [:NAT,NAT:] such that
    C4: d = {dw, db} & color dw = 0 & color db = 1 by C3, Domino;
    C5: w = dw or w = db by C4, C2, TARSKI:def 2;
          reconsider y = db as set;
  take y;
        y∈d by C4, TARSKI:def 2; then y∈union z by C2, TARSKI:def 4;
  hence y∈B by C4;
  thus [x,y]∈Z by C1, C2, C4, C5;
end;
hereby :: Z is onto B
  let y be set; assume y∈B; then
        consider b such that
    C1: y = b & b∈union z & color b = 1;
        consider d being set such that
    C2: b∈d & d∈z by C1, TARSKI:def 4;
    C3: d is domino-on-grid by A1, C2, Part cover Def;
        consider dw, db being Element of [:NAT,NAT:] such that
    C4: d = {dw, db} & color dw = 0 & color db = 1 by C3, Domino;
    C5: b = dw or b = db by C2, C4, TARSKI:def 2;
          reconsider x = dw as set;
  take x;
        x∈d by C4, TARSKI:def 2; then x∈union z by C2, TARSKI:def 4;
  hence x∈W by C4;
  thus [x,y]∈Z by C1, C2, C4, C5;
end;
let x,y,t,u be set; assume :: Z is functional
  B1: [x,y]∈Z & [t,u]∈Z;
      consider xw, yb being Element of [:NAT,NAT:] such that
  B2: [x,y] = [xw,yb] & {xw, yb}∈z & color xw = 0 by B1;
      consider tw, ub being Element of [:NAT,NAT:] such that
  B3: [t,u] = [tw,ub] & {tw, ub}∈z & color tw = 0 by B1;
  B4: x = xw & y = yb & t = tw & u = ub by B2, B3, ZFMISC 1:33;
  B5: x∈{x,y} & y∈{x,y} & t∈{t,u} & u∈{t,u} by TARSKI:def 2;
per cases by A1, B2, B3, Part cover Def;
suppose {xw,yb} = {tw,ub}; then
    (x=t or x=u) & (y=t or y=u)
    & (t=x or t=y) & (u=x or u=y) by B4, B5, TARSKI:def 2;
  hence x = t iff y = u;
suppose {xw,yb}∩{tw,ub} = Œ; then

{xw,yb} misses {tw,ub} by BOOLE:118;

       hence x = t iff y = u by B4, B5, BOOLE:def 5;
    end;
  thus thesis by A2, CARD 1:21;
end;
:::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

     set B0 = {u : u ∈ Board & color u = 0};
     set B1 = {u : u ∈ Board & color u = 1};
     set MB0 = {u : u ∈ mutilated Board & color u = 0};
     set MB1 = {u : u ∈ mutilated Board & color u = 1};

Board 01: :::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

B0 ß B1 proof

  take Z = {[w,b] where w, b is Element of Board
                  : color w = 0 & b=[w‘1,w‘2+1 mod 8]};

13 P. Rudnicki

hereby let x be set; assume x∈B0; then
          consider w such that
    B1: x = w & w∈Board & color w = 0;
          set b = [w‘1, w‘2+1 mod 8];
    B2: b‘1 = w‘1 & b‘2 = w‘2+1 mod 8 by MCART 1:7;
    B3: w‘1∈Segm 8 & w‘2∈Segm 8 by B1, MCART 1:10;
            w‘2+1 mod 8 < 8 by NAT 1:46; then
            w‘2+1 mod 8∈Segm 8 by GR CY 1:10; then
        reconsider b as Element of Board by B2, B3, MCART 1:11;
        reconsider y = b as set;
  take y;
    B4: w‘1+w‘2 mod 2 = 0 by B1, color Def;
           now assume w‘1+(w‘2+1 mod 8) mod 2 = 0; then
                 w‘2 mod 2 = (w‘2+1 mod 8) mod 2 by B4, Mod 1
                         .= (w‘2+1 mod 4 · 2) mod 2
                         .= w‘2+1 mod 2 by Mod 2; then
                 w‘2+0 mod 2=w‘2+1 mod 2; then 0 mod 2=1 mod 2 by Mod 1;
             hence contradiction by Aux 1, Mod 3;
           end; then b‘1+b‘2 mod 2 = 1 by B2, GROUP 4:100; then
        color b = 1 by color Def;
  hence y∈B1 & [x,y]∈Z by B1;
end;
hereby let y be set; assume y∈B1; then
          consider b such that
    B1: y = b & b∈Board & color b = 1;
          set w = [b‘1, b‘2+7 mod 8];
    B2: w‘1 = b‘1 & w‘2 = b‘2+7 mod 8 by MCART 1:7;
    B3: b‘1∈Segm 8 & b‘2∈Segm 8 by B1, MCART 1:10;
            b‘2+7 mod 8 < 8 by NAT 1:46; then
            b‘2+7 mod 8∈Segm 8 by GR CY 1:10; then
        reconsider w as Element of Board by B2, B3, MCART 1:11;
        reconsider x = w as set;
  take x;
    B4: b‘1+b‘2 mod 2 = 1 by B1, color Def;
           now assume
               C1: b‘1+(b‘2+7 mod 8) mod 2 = 1;
                   b‘2+0 mod 2 = b‘2 mod 2
                             .= (b‘2+7 mod 8) mod 2 by C1, B4, Mod 1
                             .= (b‘2+7 mod 4 · 2) mod 2
                             .= b‘2+7 mod 2 by Mod 2; then
                   0 mod 2 = 7 mod 2 by Mod 1 .= 6+1 mod 2
                         .= 3 · 2+1 mod 2 .= 1 mod 2 by GR CY 1:1;
             hence contradiction by Aux 1, Mod 3;
           end; then w‘1+w‘2 mod 2 = 0 by B2, GROUP 4:100; then
    B5: color w = 0 by color Def;
  hence x∈B0;
    B6: b‘2 < 8 by B3, GR CY 1:10;
           w‘2+1 mod 8 = (b‘2+7 mod 8)+1 mod 8 by MCART 1:7
                      .= b‘2+7+1 mod 8 by GR CY 1:2
                      .= b‘2+(7+1) mod 8 by AXIOMS:13
                      .= b‘2+8 mod 8 .= b‘2+1 · 8 mod 8
                      .= b‘2 mod 8 by GR CY 1:1 .= b‘2 by B6, GR CY 1:4; then
           [w‘1, w‘2+1 mod 8] = [b‘1, b‘2] by MCART 1:7 .= b by MCART 1:23;
  hence [x,y]∈Z by B1, B5;
end;

Mutilated Checkerboard in Mizar 14

  let x, y, z, u be set; assume
    A1: [x,y]∈Z & [z,u]∈Z;
          consider xw, yb being Element of Board such that
    A2: [x,y] = [xw,yb] & color xw = 0 & yb = [xw‘1, xw‘2+1 mod 8] by A1;
          consider zw, ub being Element of Board such that
    A3: [z,u] = [zw,ub] & color zw = 0 & ub = [zw‘1, zw‘2+1 mod 8] by A1;
    A4: x = xw & y = yb & z = zw & u = ub by A2, A3, ZFMISC 1:33;
  thus x=z implies y=u by A2, A3, A4;
             xw‘2∈Segm 8 & zw‘2∈Segm 8 by MCART 1:10; then
    A5: xw‘2 < 8 & zw‘2 < 8 by GR CY 1:10;
    A6: xw = [xw‘1, xw‘2] & zw = [zw‘1, zw‘2] by MCART 1:23;
  assume y=u; then
    A7: xw‘1 = zw‘1 & xw‘2+1 mod 8 = zw‘2+1 mod 8 by A2, A3, A4, ZFMISC 1:33;
              then xw‘2 mod 8 = zw‘2 mod 8 by Mod 1;
  hence x=z by A4, A6, A7, A5, Mod 3;
end;

Corners 0: ::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::
Corners c= B0 proof
      set x0 = [0,0]; set x7 = [7,7];
        0∈Segm 8 & 7∈Segm 8 by GR CY 1:10; then
      reconsider x0, x7 as Element of Board by ZFMISC 1:106;
  let x be set; assume x∈Corners; then
    A1: x=x0 or x=x7 by TARSKI:def 2; x0‘1 = 0 & x0‘2=0 by MCART 1:7; then
    A2: color x0 = 0+0 mod 2 by color Def .= 0 mod 2 .= 0 by GR CY 1:6;
        x7‘1 = 7 & x7‘2=7 by MCART 1:7; then
        color x7 = 7+7 mod 2 by color Def .= 2 · 7+0 mod 2
                .= 0 mod 2 by GR CY 1:1 .= 0 by GR CY 1:6;
  hence thesis by A1, A2;
end;

::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::
    reconsider Corners as finite set;
      Segm 8 = {k where k is Nat:k<8} by GR CY 1:def 1 .= 8 by AXIOMS:30; then
      reconsider S8 = Segm 8 as finite non empty Subset of NAT by CARD 1:69;
      [:S8, S8:] is finite; then
    reconsider Board as finite set;
      B0 c= Board proof let x be set; assume x∈B0; then
              consider w such that
          B1: x = w & w∈Board & color w = 0;
        thus thesis by B1;
      end; then B0 is Subset of Board; then
    reconsider B0 as finite set;
      MB0 c= Board proof let x be set; assume x∈MB0; then
              consider w such that
          B1: x = w & w∈mutilated Board & color w = 0;
        thus thesis by B1, BOOLE:def 4;
      end; then MB0 is Subset of Board; then
    reconsider MB0 as finite set;
       MB0 = B0 \ Corners proof
         hereby let x be set; assume x∈MB0; then consider b such that
             C1: x = b & b∈mutilated Board & color b = 0;
                 x∈Board by C1, BOOLE:def 4; then
             C2: x∈B0 by C1; not x∈Corners by C1, BOOLE:def 4;
           hence x∈B0 \ Corners by C2, BOOLE:def 4;

15 P. Rudnicki

         end;
         let x be set; assume x∈B0 \ Corners; then
             B1: x∈B0 & not x∈Corners by BOOLE:def 4; then
                 consider b such that
             B2: x = b & b∈Board & color b = 0;
                 b∈mutilated Board by B1, B2, BOOLE:def 4;
         hence x∈MB0 by B2;
       end; then
A1: card MB0 = card B0 - card Corners by Corners 0, CARD 2:63;
       now
         assume card MB0 = card B0; then card Corners = 0 by A1, REAL 2:1;
         hence contradiction by CARD 2:59;
       end; then
A2: not MB0 ß B0 by CARD 1:21;
       MB1 = B1 proof
         hereby let x be set; assume x∈MB1; then consider b such that
             C1: x = b & b∈mutilated Board & color b = 1;
                 x ∈ Board by C1, BOOLE:def 4;
           hence x∈B1 by C1;
         end;
         let x be set; assume x∈B1; then consider b such that
             B1: x = b & b∈Board & color b = 1;
                    now assume b∈B0; then consider w such that
                        C1: w = b & w∈Board & color w = 0;
                      thus contradiction by B1, C1;
                    end; then
                 not b∈Corners by Corners 0; then
                 b∈mutilated Board by B1, BOOLE:def 4;
         hence x∈MB1 by B1;
       end; then
    not MB0 ß MB1 by Board 01, A2, WELLORD2:22; then

Mutilated Board: :::::::::::::::::::::::::::::::::::::::::::::::::::::::::::
     Card {u : u ∈ mutilated Board & color u = 0}
     <> Card {v : v ∈ mutilated Board & color v = 1} by CARD 1:21;

theorem
       not ex z st z is partial-covering & union z = mutilated Board
proof
       given z such that
A1: z is partial-covering & union z = mutilated Board;

A2: Card {u : u ∈ union z & color u = 0}
    = Card {v : v ∈ union z & color v = 1} by A1, Part cover;

A3: Card {u : u ∈ mutilated Board & color u = 0}
    <> Card {v : v ∈ mutilated Board & color v = 1} by Mutilated Board;

 thus contradiction by A1, A2, A3;
end;